/*

Yanna Krupnikov	
Proposal Title: The Partisan Gender Gap: Genuine Attachment or Social Motivation?	



HYPOTHESES

Stated-Hyp1: Partisan gender gap increases when individuals are asked to report how they think vs. how they feel about partisan identity. 

	Test-Hyp1: There will be a positive interaction between gender (woman=1) and think condition (vs. feel) in terms of Democratic partisan identity

	Test-Hyp2: There will be a positive interaction between gender (woman=1) and think condition (vs. feel) in terms of strength of partisan identity
	
********************************************************************************
NOTES:

-	This is a question wording experiment. There are two experimental conditions 
and they map onto two outcomes. Not clear if the second one is supposed to be a 
moderator or what a moderator in this case would even mean. We have treated them 
as two separate experiments, which is the simplest way to test if the treatment 
in each case had any effect on the immediately related outcome. 

This also aligns with the discussion on proposal page 2.

*/

clear all
use KrupnikovS34.dta, clear

********************************************************************************


*MEASURES OF EXPERIMENTAL MANIPULATION

* think vs. feel in partisanship question
	tab FEEL_THINK
	
	/*Generally speaking, do you usually [think of yourself as/feel that you are] a Republican, a Democrat, an Independent, or what?
	*/
	recode FEEL_THINK (1/2=1) (3/4=0), gen(think_partisanship)
	tab think_partisanship // first question had think (vs. feel) condition

	/*How important, do you [feel/think], is being [a Democrat/an Independent/a Republican] to you?
	*/
	recode FEEL_THINK (1=1) (2/3=0) (4=1), gen(think_id_importance)
	tab think_id_importance // second question had think (vs. feel) condition
	
	

* OUTCOMES
	/*Generally speaking, do you usually [think of yourself as/feel that you are] a Republican, a Democrat, an Independent, or what?
	*/
	tab Q5
	replace Q5=. if Q5>7
	gen dem_identity=8-Q5 // recode so higher values indicate Democratic identity
	tab dem_identity
	
	/*How important, do you [feel/think], is being [a Democrat/an Independent/a Republican] to you?
	*/
	tab Q6
	replace Q6=. if Q6>5
	gen dem_idimp=6-Q6 // recode so higher values indicate more importance
	tab dem_idimp	
	
* MODERATORS
	
	tab GENDER
	recode GENDER (1=0) (2=1), gen(woman)
	tab woman

********************************************************************************

* ANALYSIS

*Test-Hyp1: There will be a positive interaction between gender (woman=1) and think condition (vs. feel) in terms of Democratic partisan identity

	reg dem_identity i.woman##i.think_partisanship
	// reject. p=0.124
	tess 1.woman#1.think_partisanship +, init(KrupnikovS34)
	
*Test-Hyp2: There will be a positive interaction between gender (woman=1) and think condition (vs. feel) in terms of strength of partisan identity
	
	reg dem_idimp i.woman##i.think_id_importance
	// reject. p=0.290
	tess 1.woman#1.think_id_importance +
